

UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

A computer program to convert raw Pb-U-Th isotope ratios to
blank-corrected isotope ratios and concentrations, with
associated errors and error-correlations

by

K. R. Ludwig

Open-File Report 82-820

1982

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey. Any use of trade names is for descriptive purposes only, and does not imply endorsement by the U.S. Geological Survey

INTRODUCTION

This program converts raw Pb, U, and Th isotope ratios (together with appropriate samples, spike, and aliquot weights) to blank- and discrimination-corrected isotope ratios, calculates the errors and error-correlations of these ratios, and calculates all pertinent apparent ages. The Pb spike used is assumed to be a ^{208}Pb spike, either pure or as a mixed Pb-U-Th spike, and the U and Th spikes are assumed to be ^{235}U and ^{230}Th spikes. Error-equations follow those given by Ludwig (1980), with the addition of ^{208}Pb error-equations, and take into account uncertainties in the raw isotope ratios, mass-dependent fractionation, blank amount and isotopic composition, and initial-Pb composition.

EQUIPMENT REQUIRED

The program is available in two versions: one for an HP-9830 or HP-9831 computer with at least 8K words of memory, and one for an HP-9835A or HP-9845 computer. The examples will refer to the 9835/45 version. The select code of the printer is specified in line 110 of the 9830/9831 version, and in line 260 of the 9835/45 version.

DATA REQUIRED

Before running the program, the user must store the data for the particular spikes in use, for the mass-discrimination to be applied, and for the blank amounts and ratios. These data reside in DATA lines towards the beginning of the program (appendix I), as indicated below. All uncertainties must be expressed at the 2-sigma or 95%-confidence level, in either percent or in absolute values as indicated.

(9830/45)

Line #	# Values	Significance of Values
40/330	3	1) code indicating separate (0) or mixed (1) Pb-U ((-Th) spike; 2) discrimination for Pb runs, in percent/a.m.u.; ditto for U-and Th-runs.
160/350	4	Decay constants for ^{235}U , ^{238}U , and ^{232}Th (decays/atom/million years); true $^{238}\text{U}/^{235}\text{U}$ ratio of natural uranium.
180/380	5	Pb-spike data- 1) $^{206}\text{Pb}/^{204}\text{Pb}$; 2) $^{206}\text{Pb}/^{207}\text{Pb}$; 3) $^{206}\text{Pb}/^{208}\text{Pb}$; 4) moles $^{208}\text{Pb}/\text{gram}$, 5) random uncertainty in 4), in percent 2-sigma (should be assigned a value of zero for samples spiked as a suite, but reflect the true uncertainty in the spike concentration for individual samples or samples to be compared with other laboratories).
200/420	6	U-Th spike data- 1) $^{235}\text{U}/^{238}\text{U}$; 2) moles $^{235}\text{U}/\text{gram}$; 3) random uncertainty in 2), in percent 2-sigma (should be zero for samples spiked as a suite); 4) uncertainty in $^{238}\text{U}/^{235}\text{U}$ ratios not accounted for by within-run statistics (should arise mostly from uncontrolled mass-discrimination); 5) $^{230}\text{Th}/^{232}\text{Th}$; 6) moles $^{230}\text{Th}/\text{gram}$.
220/450	6	Blank-Pb: 1) $^{206}\text{Pb}/^{204}\text{Pb}$; 2) $^{207}\text{Pb}/^{204}\text{Pb}$; 3) $^{208}\text{Pb}/^{204}\text{Pb}$; 4) variation in $^{206}\text{Pb}/^{204}\text{Pb}$; 5) variation in $^{207}\text{Pb}/^{204}\text{Pb}$; 6) variation in $^{208}\text{Pb}/^{204}\text{Pb}$. NOTE that 4), 5), and 6) are to be expressed as <u>absolute</u> 2-sigma ranges, <u>not</u> as <u>percent</u> 2-sigma ranges.
240/480	3	Blank-Pb: 1) error-correlation between $^{206}\text{Pb}/^{204}\text{Pb}$ $^{207}\text{Pb}/^{204}\text{Pb}$; 2) error correlation between $^{206}\text{Pb}/^{204}\text{Pb}$ and $^{208}\text{Pb}/^{204}\text{Pb}$; uncertainty in amount in amount of assigned blank, in percent.
260/510	4	Errors in raw Pb-isotope ratios: 1) uncertainty in discrimination, in terms of the actual range of %/a.m.u. discrimination. Thus if the discrimination ranged from about 0.03%/a.m.u. to 0.13%/a.m.u., the value would be about 0.05%. 2) uncertainty in $^{206}\text{Pb}/^{204}\text{Pb}$ ratios <u>unrelated</u> to either within-run statistics or mass-discrimination (possibly from resistor nonlinearity or other uncontrolled factors); 3) ditto for $^{206}\text{Pb}/^{207}\text{Pb}$; 4) ditto for $^{206}\text{Pb}/^{208}\text{Pb}$.

RUNNING THE PROGRAM

Change any data in the DATA lines (described above) and re-store the lines, as necessary. Make sure the printer code (line 110 in the 9830/9831 version, line 260 in the 9835/9845 version) is correct. Execute the RUN command.

<u>Query</u>	<u>Response¹/</u>
SAMPLE NAME?	Enter an appropriate name.
GRAMS OF SAMPLE, TOTAL-ALIQUOT?	Enter 2 values: the weight, in grams, of the sample in solid form, and the weight, in grams, of the total solution of the sample.
GRAMS OF UNSPIKED ALIQUOT, SPIKED ALIQUOT?	Enter 2 values: the weight of the Pb- unspiked aliquot of the total solution, and the weight of the Pb-spiked aliquot of the total solution. If all of the sample is represented by these 2 aliquots, their sum should be the total aliquot weight. If the U and Th spikes were added to the total sample (or total aliquot) rather than the Pb-spiked aliquot, enter the Pb-spiked aliquot weight as its <u>negative</u> value.
GRAMS SPIKED OF: 208, 235, 232?	Enter 3 values, indicating the amounts of Pb, U, and Th spikes added. The Pb spike must have been added to the "Pb-spiked" aliquot, whereas the U and Th spikes can have been added to either the Pb-spiked aliquot OR the total sample. The U and Th spikes must have been added together, however. If the U and Th spikes are a mixed U-Th spike, their "grams spiked" value should be the same.
UNSPIKED ALIQUOT: 206/204, %err, 206/207, %err, 206/208, %err	Enter 6 values, for the observed (uncorrected for blank and mass-discrimination) Pb-isotope ratios of the Pb-

¹ Unknown or irrelevant values should be entered as zero.

unspiked aliquot: $^{206}\text{Pb}/^{204}\text{Pb}$, error in $^{206}/^{204}$, $^{206}\text{Pb}/^{204}\text{Pb}$, error in $^{206}/^{207}$, $^{206}\text{Pb}/^{208}\text{Pb}$, percent error in $^{206}/^{208}$. The errors must be as percent 2-sigma of the mean ratio, from the mass-spectrometer run statistics only. If no good data for the Pb-unspiked aliquot are available, however, and $^{206}\text{Pb}/^{204}\text{Pb}$ ratios were taken for the Pb-spiked aliquot, enter zeroes for the first 4 values, and enter the estimated $^{206}\text{Pb}/^{208}\text{Pb}$ of the Pb-unspiked aliquot as the fifth value. The sixth value should then be the percent uncertainty of your estimate. Adequate blank- and spike-corrected Pb ratios and concentrations can generally be calculated from this information.

SPIKED-ALiquot: $^{206}/^{204}$, %err, $^{206}/^{207}$, %err, $^{206}/^{208}$, %err?

Enter 6 values for Pb-isotope ratios of the Pb-spiked aliquot, in a similar format as the previous response. Enter zeroes if no $^{206}\text{Pb}/^{204}\text{Pb}$ data were taken for the Pb-spiked aliquot, and zeroes for the third and fourth values if no $^{206}\text{Pb}/^{207}\text{Pb}$ data were taken for Pb-spiked aliquot.

SPIKED ALiquot: $^{238}/^{235}$, %err, $^{232}/^{230}$, %err?

Enter 4 values for the U-Th spiked aliquot: $^{238}\text{U}/^{235}\text{U}$ ratio (uncorrected for mass-discrimination) % error (2-sigma mean) of the ratio, $^{232}\text{Th}/^{230}\text{Th}$ ratio, % error. If no data are available for a ratio, enter zero as its value.

BLANKS (in nanograms): Pb, U, Th?

Enter 3 values indicating the estimated blanks for Pb, U, and Th, respectively, in nanograms. The Pb blank should pertain to the Pb-unspiked aliquot, whereas the U and Th blanks should pertain to the U-Th spiked aliquot.

INITIAL Pb: $^{206}/^{204}$, ERR, $^{207}/^{204}$, ERR, $^{208}/^{204}$, ERR?

Enter 6 values for the initial-Pb isotope ratios (for calculation of radiogenic-Pb ratios): $^{206}\text{Pb}/^{204}\text{Pb}$, error in $^{206}/^{204}$, $^{207}\text{Pb}/^{204}\text{Pb}$, error in $^{207}/^{204}$, $^{208}\text{Pb}/^{204}\text{Pb}$, error in $^{208}/^{204}$. If calculation of radiogenic-Pb ratios is inappropriate (e.g. for a common-Pb analysis) or not desired, enter zeroes for these values. Note that the errors for these ratios are to be expressed as absolute 2-sigma

uncertainties, not as percent uncertainties. For a suite of samples that may be assumed to have all been 'born' with the same initial-Pb, these errors should be assigned values of zero, since any errors in the assigned initial-Pb ratios will bias the whole suite of samples, but tend to cause little dispersion in the radiogenic-isotope ratios from sample to sample. The uncertainty in results for a suite of samples would then have to be evaluated by assessing the effect of biased initial-Pb ratios on the suite as a whole.

INITIAL-Pb: ERROR-CORRELATION FOR
206/204-207/204?

Enter 1 value, indicating the error-correlation for the errors in assigned initial-Pb $^{206}\text{Pb}/^{204}\text{Pb}$ and $^{207}\text{Pb}/^{204}\text{Pb}$. The true value is generally close to 1 ($\sim 0.7-0.95$), and can be calculated from a linear regression of $^{207}\text{Pb}/^{204}\text{Pb}$ on $^{206}\text{Pb}/^{204}\text{Pb}$ of appropriate candidates for the initial-Pb. If the errors of the assigned initial-Pb were zero, however, this number is irrelevant.

The values entered by query/response are stored as a 1 by 27 array ($E[1,j]$), so that with only slight modifications, the program could be used to batch-reduce data from a data file of a number of raw-data sets in the array-form $E[i,j]$, where i would be the data-set number.

EXPLANATION OF PRINTOUT

The isotope ratios labeled as "Pb-UNSPIKED AL., CORR. for $0.20 \pm 0.12 \text{ nG Pb-BLANK}$, and $0.14 \pm 0.05 \text{ \%/A.M.U. DISCRIMINATION}$ " (Fig. 1) are the blank- and discrimination-corrected ratios for the Pb-unspiked aliquot. The uncertainties below the ratios are in percent 1-sigma, and reflect the effect of uncertainties in the raw ratios, blank-Pb amount and ratios mass-discrimination, and any random run-to-run mass-spectrometer non-statistical error in excess of internal statistics. The "ERR-CORRELS:" values are the

error-correlations (ρ) between the $^{206}\text{Pb}/^{204}\text{Pb}$ - $^{207}\text{Pb}/^{204}\text{Pb}$ errors, and also the $^{206}\text{Pb}/^{204}\text{Pb}$ - $^{208}\text{Pb}/^{204}\text{Pb}$ errors. These values are essential for York-type regressions.

The value indicated as "nG COMMON Pb/UNSPKD AL." is the nanograms of common-Pb (assuming the initial-Pb isotope ratios) present in the Pb-unspiked aliquot, after correction for blank-Pb. The following value in parentheses indicates the ppm of common-Pb in the sample. These values appear only if the data indicate a significant radiogenic-Pb component to the data.

The "PPM TOTAL Pb" value is simply the Pb-content of the sample. The "EXTR. BLNK of Pb-SPIKED AL." (appears only if the $^{206}\text{Pb}/^{207}\text{Pb}$ of the Pb-spiked aliquot differs significantly from that of the Pb-unspiked aliquot) is the amount of blank-Pb in (nanograms) necessary to account for the difference in $^{206}\text{Pb}/^{207}\text{Pb}$ of the Pb-unspiked and Pb-spiked aliquots. This value should represent the blank associated with extraction and loading of the Pb-spiked aliquot, and is used for correction of the Pb-concentration values.

The "RHO(U/4-6/4)" value is the error-correlation between $^{238}\text{U}/^{204}\text{Pb}$ and $^{206}\text{Pb}/^{204}\text{Pb}$; "RHO(U/4-7/4)" is the error-correlation between $^{235}\text{U}/^{204}\text{Pb}$ and $^{207}\text{Pb}/^{204}\text{Pb}$; and "RHO(Th/4-8/4)" is the error-correlation between $^{232}\text{Th}/^{204}\text{Pb}$ and $^{208}\text{Pb}/^{204}\text{Pb}$.

The " $^{206}\text{Pb}^*$ ", " $^{207}\text{Pb}^*$ ", and " $^{208}\text{Pb}^*$ " refer to radiogenic-Pb isotopes, as do the ratio values for " $^{206}\text{Pb}/^{238}\text{U}$ " ($=^{206}\text{Pb}^*/^{238}\text{U}$), " $^{207}\text{Pb}/^{235}\text{U}$ " ($=^{207}\text{Pb}^*/^{235}\text{U}$), " $^{207}\text{Pb}/^{206}\text{Pb}$ " ($=^{207}\text{Pb}^*/^{206}\text{Pb}^*$), and " $^{208}\text{Pb}/^{232}\text{Th}$ " ($=^{208}\text{Pb}^*/^{232}\text{Th}$). The ages calculated for these values are in millions of years, and assume secular equilibrium of the radioactive daughters for the life of the systems. If the ages are very young (<10 m.y.), violations of this assumption become significant, and the ages should be recalculated following Ludwig (1977) or Ludwig (1979). Errors are expressed as percent 2-sigma. Note that the

" $^{207}\text{Pb}/^{206}\text{Pb}$ " error refers to the $^{207}\text{Pb}^*/^{206}\text{Pb}^*$ ratio, not the age. The
"RHO($^{207}\text{Pb}/^{235}\text{U}$ - $^{206}\text{Pb}/^{238}\text{U}$)" value is the error-correlation between
 $^{207}\text{Pb}^*/^{235}\text{U}$ and $^{206}\text{Pb}^*/^{238}\text{U}$.

REFERENCES

- Ludwig, K. R., 1977, Effect of initial radioactive-daughter disequilibrium on U-Pb isotope apparent ages of young minerals: U.S. Geol. Survey Jour. of Research, v. 5, p. 663-667.
- Ludwig, K. R., 1979, Solutions for apparent ages and U-pb isotope ratios of young systems without initial secular equilibrium: a computer program in Hewlett-Packard BASIC: U.S. Geol. Survey Open File Report 79-722, 15 p.
- Ludwig, K. R., 1980, Calculation of uncertainties of U-Pb isotope data: Earth Planetary Science Letters, v. 46, p. 212-220.

SAMPLE: FIGURE 1: EXAMPLE OF PRINTOUT

	206/204	% ERR	206/207	% ERR	206/208	% ERR
UNSPKD	13310	2.42	10.209	.023	15.468	.098
SPIKED			10.175	.034	1.0848	.029
SPIKE	16.2		.3128		.0000931	

BLANK: 6/4= 18.9 +/-2 7/4= 15.55 +/--.2 8/4= 37.9 +/-1.8

.07948 G Pb- 208 SPIKED, @9.7922E-09 +/- 0.00% MOLES 208/G
 .21333 G U- 235 SPIKED, @5.9800E-08 +/- .05% MOLES 235/G
 .07966 G Th- 230 SPIKED, @7.9810E-08 +/- .05% MOLES 232/G

SAMPLE WT.	TOTAL AL.	Pb-UNSPKD AL.	Pb-SPKD AL.
.00298	1.89637	1.30399	-.59238

Pb-UNSPIKED AL., CORRECTED FOR .45 +/- .27 nG Pb-BLANK,
 AND .14+/--.05 % per a.m.u. MASS-DISCRIMINATION

206/207	206/208	206/204	207/204	208/204
10.216	15.555	16644	1629.2	1070
+/--.138%	+/--.533%	+/-15.4%	+/-15.2%	+/-14.9%

ERR-CORRELS: 207/204-206/204 = .99999 208/204-206/204 = .99993

MOLES Pb PER GRAM:

206	207	208	204	TOTAL
9.72425E-07	9.51873E-08	6.25134E-08	5.84262E-11	1.13018E-06
+/- .11%				

1.72 nG COMMON Pb/UNSPKD AL. ----- (.838 PPM)
 PPM TOTAL Pb = 233.01 EXTR. BLNK OF Pb-SPKD AL. = .28 nG

238/235= 1.2365 +/--.055% 232/230= .7782 +/--.26%
 nG BLANKS: U= .15 Th= .25 U-Th SPIKED-AL. = 1.89637 GRAMS

MOLES PER GRAM:	238	235	232
	5.38775E-06	3.90756E-08	1.66709E-06
1292 PPM URANIUM		(+/--.41%)	
386.8 PPM THORIUM		(+/--.569%)	Th/U= .299

238/204= 92210 235/204= 668.8 (+/-15.4%)
 232/204 = 28530 (+/-15.4%)

RHO(U/4-6/4)= .99958 RHO(U/4-7/4)= .94161 RHO(Th/4-8/4)= .94069

MOLES PER GRAM	206*	207*	208*
	9.7149E-07	9.4291E-08	6.0361E-08

INITIAL-Pb: 6/4= 16.01 +/- .04 7/4= 15.35 +/- .015
 8/4= 36.84 +/- .08 RHO(6/4-7/4) = .92

RATIOS:	206/238	207/235	207/206	208/232
	.18031	2.413	.097058	.036207
ERRORS:	+/--.423%	+/--.427%	+/--.0614%	+/--.607%
AGES (M.Y.):	1068.7	1246.5	1568.3	718.9

RHO (207/235-206/238) = .98961

APPENDIX I -- PROGRAM LIST

```

10 ! ***** PBCONC *****
20 !
30 ! PROGRAM FOR HP-9835/9845 TO REDUCE RAW Pb-U Th ISOTOPE RATIOS TO BLANK-
40 ! CORRECTED RATIOS, CONCENTRATIONS, APPARENT-AGES, ERRORS, & ERR-CORRELATIONS
50 ! ERROR & ERROR-CORREL. CALCULATIONS FOLLOW EPSL V. 46, P. 212-220, 1980
60 ! K.R. LUDWIG. U.S. GEOLOGICAL SURVEY, 1982, OPEN-FILE REPT 83-820.
70 !
80 ! ----- NOTES -----
90 !
100 ! If no Pb-unspkd data, exists, indicate the estimated, blank-corrected
110 ! 206/208 of the sample as the unspiked 206/208 value, and enter the
120 ! unspiked 206/204 and 207/204 values as zero (i.e. unknown).
130 ! If the sample was total-spiked for U & Th, enter the Pb-spiked aliquot
140 ! weight as its negative value.
150 ! Enter unknown or irrelevant values as zero.
160 ! All ratios must be entered uncorrected for mass-discrimination.
170 ! All errors are assumed to be at the 2-sigma/95%-confidence level.
180 ! For radiogenic-Pb samples spiked as a suite for concordia or isochron
190 ! Yorkfits, initial-Pb errors should be assigned values of zero, if it
200 ! can be assumed that the samples started with the same initial-Pb ratios.
210 !
220 OPTION BASE 1
230 DEFAULT ON
240 DIM E(1,27),F$(3)[3],N$(50),Pr(2)
250 DEF FNC(R,R1)=R+R1*NOT R
251 DEF FNErr$(X,D)="+/-"&VAL$(DROUND(X,D))&"%"
260 DATA "PPB","PPM","%",2,5 ! LAST 2 #S ARE PRINTER SELECT-CODE &HP-IB CODE
270 MAT READ F$,Pr
280 I=1
290 IF Pr(2) THEN PRINTER IS Pr(1),Pr(2)
300 IF Pr(2)=0 THEN PRINTER IS Pr(1)
310 PRINT
320 ! Mixed Pb/U Spike (1=YES)?, Pb-Discr, U/Th Discr (%/AMU)
330 DATA 0,0.14,0.3
340 ! Decay Constants- 235, 238, 230; Natural 238/235 ratio
350 DATA 9.8485E-04,1.55125E-04,4.9475E-05,137.88
360 ! Pb-Spike: 206/204, 206/207, 206/208, MOLES 208/GRAM, Uncertainty in
370 ! concentration (%)
380 DATA 16.2,0.3128,9.31E-05,9.7922E-09,0
390 ! U-Spike: 238/235, Moles 235/gram, Uncertainty in concentration (%);
400 ! run-to-run uncertainty in 238/235 (%; mostly from fractionation-variation)
410 ! ditto for 232/230; Th-spike: 232/230, Moles 230/gram
420 DATA 2390,5.98E-08,0.05,0.4,0.5,774,7.981E-08
430 ! BLANK-Pb: 206/204, 207/204, 208/204, Absolute (not %) uncertainty in
440 ! 6/4, 7/4, 8/4
450 DATA 18.9,15.55,37.9,2,0.20,1.8
460 ! BLANK-Pb: Rho(6/4-7/4), Rho(6/4-8/4), +/- Blank-Amount(%)
470 ! (Rho indicates error-correlation)
480 DATA 0.7,0.7,60
490 ! Absolute error in Pb mass-discrimination, per amu, in %; non mass-discrim-
500 ! ination run-to-run errors in 6/4, 6/7, 6/8
510 DATA 0.05,0,0,0
520 ! Pb ATOMIC WTS
530 DATA 203.973,205.974,206.976,207.977
540 MAT E=ZER
550 LINPUT "SAMPLE NAME?",N$(1,50)
560 INPUT "GRAMS OF SAMPLE, TOTAL-ALIQUOT?",E(I,2),E(I,3)
570 INPUT "GRAMS OF UNSPIKED-ALIQUOT, SPIKED-ALIQUOT?",E(I,4),E(I,5)
580 INPUT "GRAMS SPIKED OF: Pb-208, U-235, Th-232?",E(I,6),E(I,7),E(I,8)
590 IF E(I,6) THEN INPUT "UNSPIKED-ALIQUOT: 206/204, %err, 206/207, %err, 206/2
08, %err?",E(I,9),E(I,10),E(I,11),E(I,12),E(I,13),E(I,14)
600 IF E(I,6) THEN INPUT "SPIKED-ALIQUOT: 206/204, %err, 206/207, %err, 206/208

```

```

, %err?",E(I,15),E(I,16),E(I,17),E(I,18),E(I,19),E(I,20)
610 IF E(I,7) OR E(I,8) THEN INPUT "SPIKED-ALICUOT: 238/235, %err, 232/230,
%err?",E(I,21),E(I,22),E(I,23),E(I,24)
620 INPUT "BLANKS (in nanograms): Pb, U, Th?",E(I,25),E(I,26),E(I,27)
630 L1=T3=L2=T5=L3=T0=0
640 IF E(I,7) THEN INPUT "INITIAL-Pb: 206/204, ERR, 207/204, ERR, 208/204, ER
R?",L1,T3,L2,T5,L3,T0
650 IF L1 THEN INPUT "INITIAL-Pb: ERROR-CORRELATION FOR 206/204-207/204?",R2
660 RESTORE 330
670 READ M,H,J,L5,L8,L9,R0,R4,R7,R8,C,J1,V1,V2,V3,U2,V0,V4,V5
680 READ B4,B3,B5,F1,F2,F0,R5,R3,E9,E5,E6,E7,E8,Aw4,Aw6,Aw7,Aw8
690 E1=100*F1/B4
700 E2=100*F2/B3
710 E0=100*F0/B5
720 E3=100*T3/(L1+NOT L1)
730 E4=100*T5/(L2+NOT L2)
740 Y2=100*T0/(L3+NOT L3)
750 ! Test to determine whether Pb data is to be reduced using 6/8 ratios only,
760 ! or with both 6/7 and 6/8 ratios.
770 D0=0
780 IF ABS(E(I,11)-E(I,17))<SQR(2*E5^2+(E(I,12)*E(I,11))^2+(E(I,18)*E(I,17))^2)/
100 THEN D0=1
790 IF (E(I,17)=0) OR (E(I,9)<22) THEN D0=1
800 PRINT LIN(2);"SAMPLE: ";N$,LIN(1)
810 IF E(I,6)=0 THEN 970
820 PRINT TAB(11);"206/204 % ERR";TAB(33);"206/207 % ERR";TAB(55);"206/208
% ERR";LIN(1)
830 Nocomp=(E(I,9)*E(I,11)=0)
840 IF Nocomp THEN 870
850 PRINT "UNSPKD";TAB(10);DROUND(E(I,9),5);TAB(19);DROUND(E(I,10),3);TAB(32);DR
OUND(E(I,11),5);TAB(41);
860 PRINT DROUND(E(I,12),3);TAB(54);DROUND(E(I,13),5);TAB(65);DROUND(E(I,14),3)
870 PRINT "SPIKED";
880 IF NOT E(I,15) THEN 900
890 PRINT TAB(10);DROUND(E(I,15),5);TAB(19);DROUND(E(I,16),3);
900 PRINT TAB(32);DROUND(E(I,17),5);TAB(41);DROUND(E(I,18),3);TAB(54);DROUND(E(I
,19),5);TAB(65);DROUND(E(I,20),3)
910 PRINT "SPIKE";TAB(10);R4;TAB(32);R7;TAB(54);R8,LIN(1)
920 PRINT "BLANK: 6/4=";B4;" +/- ";VAL$(F1);" 7/4=";B3;" +/- ";VAL$(F2);" 8/
4=";B5;" +/- ";VAL$(F0);LIN(1)
930 IMAGE 3D.5D," G Pb-",4D," SPIKED, @",D.4DE," +/-",2D.2D,"% MOLES 208/G"
940 IMAGE 3D.5D," G U-",4D," SPIKED, @",D.4DE," +/-",2D.2D,"% MOLES 235/G"
950 IMAGE 3D.5D," G Th-",4D," SPIKED, @",D.4DE," +/-",2D.2D,"% MOLES 232/G"
960 IF E(I,6) THEN PRINT USING 930;E(I,6),208,C,J1
970 IF E(I,7) THEN PRINT USING 940;E(I,7),235,V2,V3
980 IF E(I,8) THEN PRINT USING 950;E(I,8),230,V5,V3
990 PRINT LIN(1);TAB(6);"SAMPLE WT.";TAB(21);"TOTAL AL.";TAB(34);"Pb-UNSPKD AL.
";
1000 PRINT TAB(51);"Pb-SPKD AL.";LIN(1)
1010 PRINT TAB(6);E(I,2);TAB(21);E(I,3);TAB(36);E(I,4);TAB(51);E(I,5)
1020 PRINT USING 1030
1030 IMAGE 80("-")
1040 IMAGE 80("*")
1050 IF (E(I,6)=0) OR (E(I,13)=0) OR (E(I,19)=0) THEN U_th_conc
1060 !
1070 ! ++++++ Pb CONCENTRATION & RATIO CALCULATIONS ++++++
1080 !
1090 REM CORRECT Pb-ISOTOPE RATIOS FOR MASS-DISCRIMINATION
1100 E(I,9)=E(I,9)*(1+2*H/100)
1110 E(I,11)=E(I,11)*(1-H/100)
1120 IF E(I,9) AND E(I,11) THEN E(I,13)=E(I,13)*(1-2*H/100)
1130 E(I,19)=E(I,19)*(1-2*H/100)
1140 E(I,15)=E(I,15)*(1+2*H/100)
1150 E(I,17)=E(I,17)*(1-H/100)
1160 B7=B4/B3 ! Blank-Pb 206/207
1170 B8=B4/B5 ! " " 206/208

```

```

1180 B=E(I,25)          | " " amount (nG)
1190 R1=ABS(E(I,5))*E(I,2)/E(I,3)
1200 R=E(I,4)*E(I,2)*(E(I,9)*E(I,11)*E(I,13)<>0)/E(I,3)
1210 B10=B1=1/(B4+B4/B7+B4/B8+1) !MOLE-FRACT. 204, BLANK-Pb
1220 Awc=Awb=(Aw4+B4*Aw6+B3*Aw7+B5*Aw8)/(1+B4+B3+B5) !AT. WT. BLANK-Pb
1230 IF L1 THEN B10=1/(L1+L2+L3+1) !MOLE-FRACT. 204, INIT.-Pb
1240 IF L1 THEN Awc=(Aw4+L1*Aw6+L2*Aw7+L3*Aw8)/(1+L1+L2+L3) !AT. WT. INIT.-Pb
1250 Z4=1E-9*B*B1/Awb ! MOLES BLANK-204
1260 Z6=B4*Z4 ! " " 206
1270 Z7=Z6/B7 ! " " 207
1280 Z8=Z6/B8 ! " " 208
1290 G8=C*E(I,6) ! MOLES SPIKE 208
1300 G6=G8*R8 ! " " 206
1310 G7=G6/R7 ! " " 207
1320 G4=G6/R4 ! " " 204
1330 IF (E(I,9)=0) OR (E(I,11)=0) OR (D0=1) THEN 1550
1340 ! Pb concentrations and blank-corrections - using 206/207 & 206/208 of both
      spiked and unspiked aliquots
1360 D4=E(I,9) ! TRIAL BLANK-CORR. 206/204
1370 D7=E(I,11) ! " " " 206/207
1380 D8=E(I,13) ! " " " 206/208
1390 W7=(E(I,17)*G7-G6)/(1-E(I,17)/D7)-(E(I,19)*G8-G6)/(1-E(I,19)/D8)
1400 W8=(E(I,19)/B8-1)/(1-E(I,19)/D8)-(E(I,17)/B7-1)/(1-E(I,17)/D7)
1410 Z=W7/W8
1420 S5=(E(I,19)*(Z/B8+G8)-G6-Z)/(1-E(I,19)/D8)*R/R1 ! Moles sample-206/unspkd
1430 I4=D4
1440 D4=S5/((S5+Z6)/E(I,9)-Z4) !IMPROVED BLANK-CORR. 206/204
1450 D7=S5/((S5+Z6)/E(I,11)-Z7) ! " " " 206/207
1460 D8=S5/((S5+Z6)/E(I,13)-Z8) ! " " " 206/208
1470 IF I4>0 THEN 1510
1480 PRINT "*** Pb BLANK MUST BE <";B;"NG ***";LIN(1)
1490 B=2*B/3 ! Reduce blank if accounts for more than observed 204
1500 GOTO 1250
1510 IF ABS(D4-I4)>>I4/1E9 THEN 1390
1520 Z=Awb*1E9*Z/(B4*B1)
1530 GOTO Pberrors
1540 ! Pb concentrations & blank-corrections- using spiked-run data only
1550 IF (Nocomp=0) OR E(I,13) THEN 1580
1560 PRINT LIN(2);"**** EST. UNSPKD 206/208 MUST BE IN COMP 6/8 POSITION ****"
      ;LIN(3)
1570 GOTO 560
1580 IF Nocomp=0 THEN 1620
1590 E(I,10)=E(I,16)
1600 E(I,12)=E(I,18)
1610 ! Pb concentration & blank-correction - using only 206/208 spkd-run ratios
1620 D8=E(I,13)
1630 S5=(R/R1+NOT R)*(E(I,19)*(G8+Z8)-G6-Z6)/(1-E(I,19)/D8)
1640 IF Nocomp THEN 1700
1650 D8=S5/((S5+Z6)/E(I,13)-Z8)
1660 D7=S5/((S5+Z6)/E(I,11)-Z7)
1670 D4=S5/((S5+Z6)/E(I,9)-Z4)
1680 IF D4<=0 THEN 1480
1690 GOTO Pberrors
1700 D7=S5/((S5+Z6+G6)/E(I,17)-Z7-G7)
1710 D4=S5/((S5+Z6+G6)/E(I,15)-Z4-G4)
1720 GOTO 1680
1721 ! Calculate errors & error-correlations for blank-corr. Pb ratios
1730 Pberrors: D=Z6/S5
1740 F=Z7*D7/S5
1750 E=Z4*D4/S5
1760 G=Z8*D8/S5
1780 Z0=E(I,14)^2+4*E5^2+E8^2
1790 Z5=E(I,20)^2+4*E5^2+E8^2
1800 W7=(E(I,19)-R8)^2*Z0+(D8-R8)^2*Z5
1810 W7=J1^2+W7*(E(I,19)/((D8-E(I,19))*(E(I,19)-R8)))^2
1820 W8=E(I,10)^2+E6^2

```

```

1830 W9=E(I,12)^2+E7^2
1840 W0=E(I,14)^2+E8^2
1850 F4=(1+E)^2*W8+(E*E1)^2+(D-E)^2*W7+((E-D)*E9)^2+((2+2*E)*E5)^2 ! 6/4 VAR%
1860 F5=(1+E)^2*W8+(1+F)^2*W9+((F-E)*E1)^2+(F*E2)^2+(F-E)^2*W7
1870 F5=F5+((E-F)*E9)^2-2*F*(F-E)*R5*E1*E2+((3+2*E+F)*E5)^2 ! 7/4 VAR%
1880 F7=(1+E)^2*W8+(1+G)^2*W0+((E-G)*E1)^2+(G*E0)^2+(E-G)^2*W7
1890 F7=F7+((E-G)*E9)^2-2*G*(G-E)*E1*E0*R3+((4+2*E+2*G)*E5)^2 ! 8/4 VAR%
1900 R9=(1+E)^2*W8-E*(F-E)*E1^2+(D-E)*(F-E)*W7+(E-D)*(E-F)*E9^2
1910 R9=(R9+(3+2*E+F)*(2+2*E)*E5^2+E*F*R5*E1*E2)/SQR(F4*F5) ! Rho 6/4-7/4
1920 R6=(1+E)^2*W8+E*(E-G)*E1^2+(D-E)*(G-E)*W7+(D-E)*(G-E)*E9^2
1930 R6=(R6+(2+2*E)*(4+2*E+2*G)*E5^2+E*G*E1*E0*R3)/SQR(F4*F7) ! Rho 6/4-8/4
1940 C1=S5/FNC(R,R1) ! MOLES 206/G
1950 Z0=(Aw6*C1+Aw7*C1/D7+Aw8*C1/D8+Aw4*C1/D4)*1E6 ! PPM Pb/G
1960 Q1=Awc*1E6*S5/(B10*D4*FNC(R,R1)) ! PPM COMMON-Pb
1970 G0=SQR(F4+F5-2*R9*SQR(F4*F5)) ! 6/7 SIG%
1980 G1=SQR(F4+F7-2*R6*SQR(F4*F7)) ! 6/8 SIG%
1990 PRINT TAB(8); "Pb-UNSPIKED AL., CORRECTED FOR"; B; "+/-"; DROUND(E9*B/100,2); "n
G Pb-BLANK,"; LIN(1); TAB(10);
2000 PRINT "AND "; VAL$(H); "+/-"; VAL$(E5); " % per a.m.u. MASS-DISCRIMINATION",LIN
(1)
2010 PRINT " 206/207"; TAB(18); "206/208"; TAB(32); "206/204"; TAB(47); "207/204"; TAB
(62); "208/204"
2020 PRINT TAB(2); DROUND(D7,5); TAB(18); DROUND(D8,5); TAB(32); DROUND(D4,5); TAB(47)
; DROUND(D4/D7,5); TAB(62); DROUND(D4/D8,5)
2030 PRINT TAB(3); FNErr$(G0,3); TAB(18); FNErr$(G1,3); TAB(32); FNErr$(SQR(F4),3); TA
B(47); FNErr$(SQR(F5),3); TAB(62); FNErr$(SQR(F7),3); LIN(1)
2050 PRINT "ERR-CORRELS: 207/204-206/204 ="; DROUND(R9,5); " 208/204-206/204 =
"; DROUND(R6,5); LIN(1)
2060 PRINT TAB(26); "MOLES Pb PER GRAM:"
2070 IMAGE 5(D.5DE,4X),/, " +/-", 2D.DD, "%", /
2080 IMAGE 4X, "206", 12X, "207", 12X, "208", 12X, "204", 11X, "TOTAL"
2090 PRINT USING 2080
2100 PRINT USING 2070; C1, C1/D7, C1/D8, C1/D4, C1+C1/D7+C1/D8+C1/D4, SQR(W7)
2110 IF D0=0 THEN PRINT DROUND(1000*Q1*FNC(R,R1),3); "nG COMMON Pb/UNSPKD AL. --
--- ("; VAL$(DROUND(Q1,3)); " PPM)"
2120 Mult=1000*(Z0<1)+1E-4*(Z0>1E4)+1*(Z0=1)*(Z0<=1E4)
2130 PRINT F$(1+(Z0>1)+(Z0>1E4)); " TOTAL Pb ="; DROUND(Z0*Mult,5);
2140 IF D0=0 THEN PRINT SPA(5); "EXTR. BLNK OF Pb-SPKD AL. ="; DROUND(Z,2); "nG";
2150 PRINT
2160 !
2170 ! ++++++ U,Th CONCENTRATIONS & ERRORS ++++++
2180 !
2190 U_th_conc: IF E(I,7)*E(I,21)=0 THEN 3300
2210 S=R1*(E(I,5)>0)+(E(I,5)<=0)*E(I,2)
2211 Ru=E(I,21)*(1+.03*J) ! FRACT.-CORR. 238/235
2220 T6=E(I,7)*V2*(Ru-1/V1)/(1-Ru/R0)-4.171E-12*E(I,26)
2221 Rth=E(I,23)*(1+.02*J) ! FRACT.-CORR. 232/230
2230 L=(Rth<0)*(E(I,8)*V5*(Rth-1/(V4+NOT V4))-E(I,27)*1E-9/232)
2240 K6=L*D4/(S*C1) ! 232/204
2250 P=(L<0)*2.32038E8*L/S ! PPM Th
2260 Q=238.03*(T6+T6/R0)/S ! G U/G
2270 REM ERRORS AND ERROR-CORRELATIONS - U-Th CONCS AND U,Th/204 RATIOS
2280 W7=W7-M*J1^2
2290 U=(R0/(R0-E(I,21)))^2*(E(I,22)^2+U2^2)
2300 T8=W8*(1+E)^2+(E*E1)^2+W7*(1+E)^2+(E*E9)^2+((2+2*E)*E5)^2+U+NOT M*V3^2
2310 T4=W8*(1+E)^2+(E*E1)^2-(1+E)*(D-E)*W7+E*(E-D)*E9^2
2320 T4=(T4+2*(1+E)*E5^2)/SQR(F4*T8)
2330 T7=W8*(1+E)^2+E*(F-E)*E1^2-(1+E)*(F-E)*W7+E*(E-F)*E9^2
2340 T7=(T7+2*(3+2*E+F)*(1+E)*E5^2-E*F*R5*E1*E2)/SQR(F5*T8)
2350 P1=SQR(V3^2+(E(I,23)/(E(I,23)-1/V4))^2*(E(I,24)^2+V0^2))
2360 T9=W8*(1+E)^2+(E*E1)^2+(1+E)^2*W7+(E*E9)^2+((2+2*E)*E5)^2+P1^2+NOT M*V3^2
2380 T2=(1+E)^2*W8+E*(G-E)*E1^2-(1+E)*(G-E)*W7+E*(E-G)*E9^2
2390 T2=(T2+(4+2*E+2*G)*(2+2*E)*E5^2-E*G*R3*E1*E0)/SQR(T9*F7)
2400 K5=T6*D4/(S*C1) ! 238/204
2410 PRINT USING 1030
2420 STANDARD

```

```

2430 PRINT "238/235=";E(I,21);" ";FNERR$(E(I,22),3);
2440 IF E(I,8)*E(I,23)=0 THEN 2460
2450 PRINT TAB(37);"232/230=";E(I,23);" ";FNERR$(E(I,24),3);
2460 PRINT LIN(1);"nG BLANKS: U=";E(I,26);TAB(22);"Th=";E(I,27);TAB(36);"U-Th"
;
2470 PRINT " SPIKED-AL. =" ;E(I,3)*(E(I,5)<=0)+E(I,5)*(E(I,5)>0);"GRAMS";LIN(1)
2480 PRINT TAB(23);"238";TAB(40);"235";TAB(57);"232"
2490 FLOAT 5
2500 PRINT "MOLES PER GRAM:";TAB(19);T6/S;TAB(36);T6/(S*R0);TAB(53);L/S,LIN(1)
2510 STANDARD
2520 Mult=1E9*(Q<1E-6)+1E6*(Q=1E-6)*(Q<.01)+100*(Q=.01)
2530 PRINT TAB(4);DROUND(Q*Mult,4);TAB(12);F$(1+(Mult<1E9)+(Mult<1E6));TAB(17);"
URANIUM";TAB(40);"("&FNERR$(SQR(U+V3^2),3)&")"
2550 IF L=0 THEN 2580
2560 Mult=1000*(P<1)+1*(P=1)*(P<1E4)+1E-4*(P=1E4)
2570 PRINT TAB(4);DROUND(P*Mult,4);TAB(12);F$(1+(Mult<1000)+(Mult<1));TAB(17);"T
HORUIM";TAB(40);"("&FNERR$(SQR(P1^2+V3^2),3);"");
2575 PRINT TAB(60);"Th/U=";DROUND(P/Q/1E6,3)
2580 PRINT LIN(1);"238/204=";DROUND(K5,4)," 235/204=";DROUND(K5/R0,4);TAB(53)
;"("&FNERR$(SQR(T8),3);"");
2590 IF L=0 THEN 2610
2600 PRINT "232/204 =" ;DROUND(K6,4);TAB(53);"("&FNERR$(SQR(T9),3)&")";LIN(1)
2610 PRINT "RHO(U/4-6/4)=";DROUND(T4,5);" RHO(U/4-7/4)=";DROUND(T7,5);
2620 IF L=0 THEN 2640
2630 PRINT " RHO(Th/4-8/4)=";DROUND(T2,5);
2640 PRINT
2650 PRINT USING 1030
2660 IF L1<5 THEN 3270
2665 !
2670 ! ++++++ CALCULATE RADIOGENIC-ISOTOPE RATIOS ++++++
2675 !
2675 !
2680 A4=C1*(D4-L1)/D4 ! Moles radiogenic Pb-206/gram
2690 A7=C1*(1/D7-L2/D4) ! " " Pb-207/gram
2700 A8=C1*(1/D8-L3/D4) ! " " Pb-208/gram
2710 R=FNC(R,R1)
2750 G1=S5*L1/(D4*A4*R)
2760 R7=(S5+Z6)/(A4*R)
2770 B3=Z6/(A4*R)
2780 G4=L2*S5/(D4*A7*R)
2790 B5=(S5/D7+Z7)/(A7*R)
2800 G6=Z7/(A7*R)
2810 G7=(S5+Z4*D4)/S5
2820 G8=D4*Z4/S5
2830 T2=L3*S5/(D4*A8*R)
2840 T4=(S5/D8+Z8)/(A8*R)
2850 T7=Z8/(A8*R)
2860 G2=0
2870 W8=W8+4*E5^2
2880 R8=W8*(G1*G7)^2+(G1*G8*E1)^2+(G1*E3)^2+W7*(R7-G1*G7)^2+((B3-G1*G8)*E9)^2+U
2890 G0=W8*(G4*G7)^2+((G6-G8*G4)*E1)^2+(G4*E4)^2+(W9+E5^2)*B5^2+(G6*E2)^2
2900 G0=G0+W7*(B5-G7*G4)^2+((G6-G8*G4)*E9)^2-2*G6*(G6-G8*G4)*R5*E1*E2+U
2910 IF (E(I,8)=0) OR (E(I,23)=0) THEN 2940
2920 G2=(T2*G7)^2*W8+((T7-T2*G8)*E1)^2+(T2*Y2)^2+(W0+4*E5^2)*T4^2+(T7*E0)^2
2930 G2=G2+(T4-T2*G7)^2*W7+((T7-T2*G8)*E9)^2-2*T7*(T7-T2*G8)*R3*E1*E0+P1^2
2940 C=W8*G1*G4*G7^2+G1*G8*(G8*G4-G6)*E1^2+(R7-G1*G7)*(B5-G4*G7)*W7
2950 C=C+(B3-G1*G8)*(G6-G4*G8)*E9^2+G1*G4*R2*E3*E4+G1*G8*G6*R5*E1*E2+U
2960 B1=R8+G0-2*C
2970 C=C/SQR(R8*G0)
2975 ! Calculate Pb/U & Pb/Th apparent ages
2980 K1=A4*S/T6 ! 206*/238
2990 K2=A7*S/R0/T6 ! 207*/235
3000 K3=A7/A4 ! 207*/206*
3010 Y8=(L<>0)*S*A8/(L+NOT L) ! 208*/232
3030 U3=(K1-1)*LOG(1+K1+2*(K1<=-1))/L8 ! 206*/238 age
3040 Y2=(K2-1)*LOG(1+K2+2*(K2<=-1))/L5 ! 207*/235 age

```

```

3050 I4=(Y8>-1)*(Y8<=.5)*LOG(1+Y8+2*((Y8<=-1)+(Y8>.5)))/L9      ! 208*/232 age
3060 ! Calculate 207*/206* age
3070 IF K3>.0156 THEN 3100
3080 Z=-50000
3090 GOTO 3160
3100 B=3000*(SGN(-1/2+(K3>L5/(L8*R0)))+(K3>.7))
3110 F=L8/L5*EXP(B*(L8-L5))
3120 Z=LOG(1+(EXP(B*L8)-1-F*(EXP(B*L5)-1))/(1/(R0*K3)-F))/L5
3130 IF ABS(Z-B)<.001 THEN 3160
3140 B=Z
3150 GOTO 3110
3160 PRINT TAB(21);"206*";TAB(37);"207*";TAB(51);"208*"
3170 FLOAT 4
3180 PRINT "MOLES PER GRAM";TAB(17);A4;TAB(33);A7;TAB(47);A8,LIN(1)
3190 STANDARD
3200 PRINT "INITIAL-Pb: 6/4=";L1;" +/- ";T3;" 7/4=";L2;" +/- ";T5;
3210 PRINT LIN(1);TAB(14);"8/4=";L3;" +/- ";T0,"RHO(6/4-7/4) =";R2,LIN(2)
3220 PRINT TAB(16);"206/238";TAB(32);"207/235";TAB(47);"207/206";TAB(61);"208/2
32",LIN(1)
3230 PRINT "RATIOS:";TAB(15);DROUND(K1,5);TAB(31);DROUND(K2,5);TAB(46);DROUND(K
3,5);TAB(60);DROUND(Y8,5)
3240 PRINT "ERRORS:";TAB(15);FNERR$(SQR(R8),3);TAB(31);FNERR$(SQR(G0),3);TAB(46
);FNERR$(SQR(B1),3);TAB(61);FNERR$(SQR(G2),3)
3250 PRINT "AGES (M.Y.):";TAB(15);DROUND(U3,5);TAB(31);DROUND(Y2,5);TAB(46);DROU
ND(Z,5);TAB(60);DROUND(I4,5)
3260 PRINT LIN(1);"RHO (207/235-206/238) =";DROUND(C,5)
3270 PRINT USING 1040
3280 PRINT LIN(2)
3290 STANDARD
3300 GOTO 540
3310 END

```